11 Ciclos limite e dinâmica populacional

Problema 3
Uma população de dragões, y, e uma população de águias, x, evoluem de acordo com um modelo de Lotka-Volterra:

$$\dot{x} = x(2-y) \qquad\qquad \dot{y} = \frac{y}{2}(x-3)$$

Analise a estabilidade e desenhe o retrato de fase do sistema. Qual será o estado limite? alguma das duas espécies será extinta?

As componentes da velocidade de fase são:

```
(%i1) u: [x*(2-y), y*(x-3)/2]$
```

e os pontos de equilíbrio são os pontos onde as duas componentes da velocidade de fase são nulas:

```
(%i2) p: solve (u);
```

(%o2) $[\,[y = 0,\ x = 0],\ [y = 2,\ x = 3]\,]$

A matriz jacobiana do sistema é:

```
(%i3) J: jacobian (u, [x,y]);
```

(%o3) $\begin{bmatrix} 2-y & -x \\ \frac{y}{2} & \frac{x-3}{2} \end{bmatrix}$

e os valores próprios da matriz da aproximação linear, na vizinhança do primeiro ponto de equilíbrio, (0, 0), são,

```
(%i4) eigenvalues (subst (p[1], J));
```

(%o4) $\left[\left[-\frac{3}{2},\ 2\right],\ [1,\ 1]\right]$

Ou seja, o ponto de equilíbrio em (0, 0) é ponto de sela. Os valores próprios da matriz da aproximação linear, na vizinhança do segundo ponto de equilíbrio, (3, 2), são:

```
(%i5) eigenvalues (subst (p[2],J));
```

(%o5) $[\,[\,-\sqrt{3}\,i\,,\ \sqrt{3}\,i\,]\,,\ [\,1\,,\ 1\,]\,]$

E, por serem números imaginários puros, o segundo ponto de equilíbrio é um centro.

O retrato de fase, na região relevante onde as duas populações x e y são positivas ou nulas, constrói-se com o seguinte comando:

```
(%i6) plotdf (u, [x,y], [x,0,10], [y,0,10]);
```



O estado limite é um ciclo, em que as populações das duas espécies oscilam, sem que nenhuma das duas seja nunca extinta.

Problema 4
Considere o modelo de Verhulst para duas populações:

$$\dot{x} = x\,(1 - x - 2\,y) \qquad\qquad \dot{y} = y\,(1 + 5\,x - y)$$

diga se é um sistema com competição ou um sistema predador presa (e nesse caso quais as presas e quais os predadores). Analise a estabilidade e desenhe o retrato de fase.

O termo $-2\,y$ na expressão de $\dot{x}$ implica que a população y faz diminuir a população x. E o termo $+5\,x$ na expressão de $\dot{y}$ implica que a população x faz aumentar a população y. Como tal, trata-se de um sistema predador presa, onde x são as presas e y os predadores.

As componentes da velocidade de fase são:

```
(%i1) u: [x*(1-x-2*y), y*(1+5*x-y)]$
```

e os pontos de equilíbrio são os pontos onde as duas componentes da velocidade de fase são nulas:

```
(%i2) p: solve (u);
```

$$(\%o2) \quad \left[[y=0,\ x=0],\ [y=0,\ x=1],\ [y=1,\ x=0],\ \left[y=\frac{6}{11},\ x=-\frac{1}{11}\right]\right]$$

Como só interessam os valores positivos das variáveis de estado, o sistema tem então 3 pontos de equilíbrio, nos pontos (0, 0), (1, 0) e (0, 1) do espaço de fase (x, y).

A matriz jacobiana do sistema é:

```
(%i3) J: jacobian (u, [x,y])$
```

As matrizes das aproximações lineares nas vizinhanças dos 3 pontos de equilíbrio são então:

```
(%i4) makelist (subst (p[i], J), i, 1, 3);
```

$$(\%o4) \quad \left[\begin{bmatrix}1 & 0\\ 0 & 1\end{bmatrix},\ \begin{bmatrix}-1 & -2\\ 0 & 6\end{bmatrix},\ \begin{bmatrix}-1 & 0\\ 5 & -1\end{bmatrix}\right]$$

A primeira matriz é diagonal e com um único valor próprio, igual a 1. Como tal, o primeiro ponto de equilíbrio, na origem do espaço de fase, é um nó próprio repulsivo.

Os valores próprios nos outros dois pontos de equilíbrio são os seguintes:

```
(%i5) makelist( eigenvalues (subst (p[i], J))[1], i, 2, 3);
```

$$(\%o5) \quad [[\ -1,\ 6\ ],\ [\ -1\ ]]$$

Ou seja, o segundo ponto de equilíbrio, (1, 0), é ponto de sela e terceiro ponto de equilíbrio, (0, 1), é um nó impróprio atrativo.

O retrato de fase, na região relevante onde as duas populações x e y são positivas ou nulas, constrói-se com o seguinte comando:

```
(%i5) plotdf (u, [x,y], [x,0,2], [y,0,2]);
```



Se inicialmente existem predadores (y maior que zero), o sistema evolui sempre até extinguirem-se todas as presas, ficando a população de predadores igual a uma unidade.

Problema 6
O sistema dinâmico:

$$\dot{x} = y + x\,(x^2 + y^2) \qquad\qquad \dot{y} = -x + y\,(x^2 + y^2)$$

tem um ponto de equilíbrio na origem. Encontre as equações de evolução em coordenadas polares, nomeadamente, as expressões para $\dot{r}$ e $\dot{\theta}$ em função de r e θ. Explique que tipo de ponto de equilíbrio é a origem e quantos ciclos limite existem.

As derivadas das expressões $x = r\,\cos\theta$ e $y = r\,\sin\theta$ são:

$$\dot{x} = \dot{r}\,\cos\theta - r\,\dot{\theta}\,\sin\theta$$
$$\dot{y} = \dot{r}\,\sin\theta + r\,\dot{\theta}\,\cos\theta$$

Substituindo nas equações de evolução, obtém-se as equações de evolução em coordenadas polares:

$$\dot{r}\cos\theta - r\dot{\theta}\sin\theta = r\sin\theta + r^3\cos\theta$$
$$\dot{r}\sin\theta + r\dot{\theta}\cos\theta = -r\cos\theta + r^3\sin\theta$$

que são duas equações lineares para $\dot{r}$ e $\dot{\theta}$. Aplicando qualquer método de resolução de equações lineares, obtém-se essas duas expressões. Por exemplo, o método de eliminação; multiplicando a primeira equação por $\cos\theta$ e a segunda por $\sin\theta$,

$$\dot{r}\cos^2\theta - r\dot{\theta}\sin\theta\cos\theta = r\sin\theta\cos\theta + r^3\cos^2\theta$$
$$\dot{r}\sin^2\theta + r\dot{\theta}\sin\theta\cos\theta = -r\sin\theta\cos\theta + r^3\sin^2\theta$$

e somando as duas equações obtêm-se a expressão para $\dot{r}$

$$\dot{r} = r^3$$

Multiplicando a primeira equação de evolução por $\sin\theta$ e a segunda por $\cos\theta$,

$$\dot{r}\sin\theta\cos\theta - r\dot{\theta}\sin^2\theta = r\sin^2\theta + r^3\sin\theta\cos\theta$$
$$\dot{r}\sin\theta\cos\theta + r\dot{\theta}\cos^2\theta = -r\cos^2\theta + r^3\sin\theta\cos\theta$$

e subtraindo a primeira equação da segunda obtêm-se a expressão para $\dot{\theta}$

$$r\dot{\theta} = -r \qquad \Longrightarrow \qquad \dot{\theta} = -1 \quad (\text{se:} r \neq 0)$$

Fora da origem, r é positiva e, como tal, $\dot{r} = r^3$ é sempre positiva. Ou seja, o estado do sistema afasta-se sempre da origem (r aumenta). Enquanto o estado se afasta da origem, dá várias voltas no sentido negativo (sentido dos ponteiros do relógio), porque $\dot{\theta}$ é igual a -1. Isso implica que a origem é um foco repulsivo e não existe nenhum ciclo limite.

As expressões para $\dot{r}$ e $\dot{\theta}$ também podem ser obtidas no Maxima com os seguintes comandos:

```
(%i1) x: r*cos(q)$

(%i2) y: r*sin(q)$

(%i3) gradef(r,t,v)$
```

```
(%i4) gradef(q,t,w)$
(%i5) e1: diff(x,t) = y+(x^2+y^2)*x;
```

(%o5) $\cos(q)\, v - \sin(q)\, r\, w = \cos(q)\, r\, \left(\sin^2(q)\, r^2 + \cos^2(q)\, r^2\right) + \sin(q)\, r$

```
(%i6) e2: diff(y,t) = -x+(x^2+y^2)*y;
```

(%o6) $\cos(q)\, r\, w + \sin(q)\, v = \sin(q)\, r\, \left(\sin^2(q)\, r^2 + \cos^2(q)\, r^2\right) - \cos(q)\, r$

```
(%i7) trigsimp(solve([e1,e2],[v,w]));
```

(%o7) $[\,[\, v = r^3 \,,\, w = -1 \,]\,]$

Problema 7
Em relação ao seguinte sistema não linear:

$$\dot{x} = x - y - x^3 - x y^2 \qquad \dot{y} = x + y - x^2 y - y^3$$

(a) Encontre as equações de evolução em coordenadas polares (sugestão: use o comando trigreduce para simplificar o resultado).

(b) Trace o gráfico de $\dot{r}$ em função de r (r não pode ser negativo), demonstre que o sistema tem um único ciclo limite e determine se é atrativo ou repulsivo.

(c) Escreva a equação do ciclo limite, em função das coordenadas cartesianas (x, y).

(d) Corrobore a resposta traçando o retrato de fase no plano cartesiano (x, y).

(a) Substituem-se as coordenadas cartesianas por coordenadas polares nas duas equações de evolução, e resolvem-se em simultâneo para encontrar as expressões para $\dot{\theta}$ e $\dot{r}$ (designadas por w e v nos comandos seguintes):

```
(%i1) [x, y]: [r*cos(q), r*sin(q)]$
(%i2) gradef (r,t,v)$
(%i3) gradef (q,t,w)$
(%i4) trigsimp (solve( [diff(x,t)=x-y-x^3-x*y^2,
          diff(y,t)=x+y-x^2*y-y^3], [v,w]));
```

```
(%o4)    [[v = r - r^3 ,  w = 1]]
```

As duas equações de evolução, em coordenadas polares, são então: $\dot{r} = r - r^3$, $\dot{\theta} = 1$.

(b) O gráfico de $\dot{r}$ em função de r obtém-se com o comando:

```
(%i5) plot2d (rhs(%[1][1]), [r,0,2]);
```



e mostra que existe uma única raiz diferente de zero, em $r = 1$, e r aumenta se for menor que 1 e diminui se for maior que 1. Assim sendo, existe um único ciclo limite, atrativo, que é uma circunferência de raio 1.

(c) O ciclo limite é a circunferência de raio 1 e centro na origem, que em coordenadas cartesianas tem equação $x^2 + y^2 = 1$

(d) Para criar o retrato de fase, em coordenadas cartesianas, é necessário eliminar primeiro a definição das coordenas polares:

```
(%i6) remvalue(x,y)$

(%i7) plotdf([x-y-x^3-x*y^2,x+y-x^2*y-y^3],[x,y],[x,-2,2],[y,-2,2],
     [vectors,""]);
```


2
1
0
-1
-2
y
-2
-1
0
1
x